Croustillant
sur lit de calcul formel

RENCONTRE D'UNE PARABOLE AVEC CERCLE



Le cercle de centre Ω coupe la parabole en quatre points : A, B, C et D.
G est l'isobarycentre de ABCD.

à faire

- Construire une figure à l'aide d'un logiciel de géométrie dynamique et émettre une conjecture concernant le lieu de G.
- Démontrer cette conjecture.

pour travailler avec Xcas, on va se placer dans un repère tel que l'équation de la parabole est de la forme $y = x^2$.

$$\begin{cases} \mathrm{M} \in \mathscr{C}_f \Leftrightarrow y_\mathrm{M} = x_\mathrm{M}^2 \\ \mathrm{M} \in \mathscr{C}(\Omega, r) \Leftrightarrow (x_\mathrm{M} - x_\Omega)^2 + (y_\mathrm{M} - y_\Omega)^2 - r^2 = 0 \end{cases}$$

Xcas ne sais pas résoudre l'équation : $(x-xw)^2+(x^2-yw)^2-r^2 = 0$ d'inconnue x... Avec des valeurs numériques pour xw, yw et r il donne les valeurs approchées des solutions.

On sait que l'intersection de la parabole avec le cercle doit donner 4 points.
On suppose l'existence de ces points de coordonnées $(a\,;a^2)$, $(b\,;b^2)$, $(c\,;c^2)$ et $(d\,;d^2)$. Puis on exprime le fait que ces points sont cocycliques.

```
/* on sait que l'intersection du cercle et de la parabole donne 4 points */
/* le cercle a pour centre W(xw,yw) et pour rayon r */
C:=(x-xw)^2 + (y-yw)^2 - r^2
/* comme il y a intersection avec la parabole, les points d'intersection */
/* ont des coordonnées de la forme (x ; x^2) */
/* les abscisses de ces 4 points sont les racines de P */
P:=unapply(subst(C,y=x^2),x)
/* P admet 4 racines : il peut s'écrire sous la forme Q(x) */
Q(x):=(x-a)*(x-b)*(x-c)*(x-d)
coeff(P(x))
coeff(Q(x))
/* on identifie les coefficients */
eq1:=coeff(P(x))[1]=coeff(Q(x))[1]
/* donc l'isobarycentre a pour abscisse 0... */
/* pour le plaisir, on cherche (xw, yw) et r en fonction de a, b et c */
eq2:=coeff(P(x))[2]=coeff(Q(x))[2]
eq3:=coeff(P(x))[3]=coeff(Q(x))[3]
eq4:=coeff(P(x))[4]=coeff(Q(x))[4]
eq22:=simplifier(subst(eq2,d=-(a+b+c)))
eq32:=simplifier(subst(eq3,d=-(a+b+c)))
eq42:=simplifier(subst(eq4,d=-(a+b+c)))
solve([eq22,eq32,eq42],[xw,yw,r^2])
/* donc on trouve : xw=-(b+c)*(a+c)*(a+b)/2 ; yw=(a*(b+a)+b*(c+b)+c*(a+c)+1)/2  */
/* et r =sqrt(((b+c)^2+1)*((a+c)^2+1)*((a+b)^2+1)/4) */
```

Croustillant
sur lit de calcul formel

RENCONTRE D'UNE PARABOLE AVEC CERCLE

pour continuer

- coordonnées des Ω, centre du cercle.
- expression de r, rayon du cercle.

Commande

coeff, simplify, solve, sqrt, subst, unapply

Croustillant
sur lit de calcul formel

RENCONTRE D'UNE PARABOLE AVEC CERCLE

Version élève

On donne la parabole et le cercle

Soit la parabole $\mathscr{P}$ d'équation $y = x^2$ et le cercle $\mathscr{C}$ de centre $\Omega(-1\,;11)$ et de rayon $5\sqrt{2}$.

Déterminer les coordonnées des points d'intersection du cercle et de la parabole. En déduire celle de l'isobarycentre.

$\mathscr{C}$ a pour équation $(x+1)^2 + (y-11)^2 = 50$

Les coordonnées des points d'intersection vérifient : $(x+1)^2 + (x^2-11)^2 - 50 = 0$

$x^4 - 21x^2 + 2x - 72 = 0$

la figure permet de conjecturer que -2 et 4 sont des solutions de l'équation.

On le vérifie en remplaçant les lettres par leur valeurs.

On peut alors factoriser : $(x+2)(x-4)(x^2+2x-9) = 0$

la somme des racines de x^2+2x-9 est -2, donc l'abscisse de G est $\dfrac{-2+4+(-2)}{4} = 0$

(Xcas sait trouver directement les racines...)

```
/* coordonnées des points d'intersection de la parabole x^2 */
/* et du cercle d'équation (x+1)^2+(y-11)^2=50 */
C:=(x+1)^2+(y-11)^2-50
Cx:=subst(C,y=x^2)
expand(Cx)
subst(C,x=-2,y=4)
subst(C,x=4,y=16)
normal(Cx/((x+2)*(x-4)))
/* on aurait pu directement demander */
solve(Cx=0,x)

```

Croustillant

sur lit de calcul formel

RENCONTRE D'UNE PARABOLE AVEC CERCLE

On donne la parabole et 3 points

Soit $\mathscr{P}$ la parabole d'équation $y = x^2$ et les points A(2 ; 4), B(−4 ; 16) et C(3 ; 9).

Calculer les coordonnées du centre du cercle $\mathscr{C}$ passant par les points A, B et C ainsi que son rayon.

En déduire les coordonnées du point D, quatrième point d'intersection de $\mathscr{P}$ et $\mathscr{C}$.

Vérifier que le point G isobarycentre des points A, B, C et D appartient à l'axe de symétrie de la parabole.

Le centre du cercle est l'intersection des médiatrices des segments [AB] et [AC]

équation de (AB) : $y = \dfrac{y_B - y_A}{x_B - x_A}(x - x_A) + y_A$

$y = \dfrac{12}{-6}(x-2) + 4 = -2x + 8$

milieu de [AB] : $M_1 = \left(\dfrac{-4+2}{2}\,;\dfrac{16+4}{2}\right) = (-1\,;10)$

donc équation de la médiatrice : $\Delta_{AB} : y = \dfrac{1}{2}(x - (-1)) + 10 = \dfrac{1}{2}x + \dfrac{21}{2}$

On trouve de même l'équation de la médiatrice Δ_{AC}, puis le point d'intersection et enfin l'équation du cercle...

Chercher l'intersection du cercle et de la parabole revient à résoudre :

$(x+5)^2 + (x^2-8)^2 = 65$ qui est une équation de degré 4 dont on connaît 3 racines (2, −4 et 3) on trouve facilement la quatrième (produit des racines : $25+64-65 = 24 = 2 \times (-4) \times 3 \times \boxed{\text{(-1)}}$ — ou identification des coefficients — ou ...)

L'abscisse de l'isobarycentre est : $\dfrac{2+(-4)+3+(-1)}{4} = 0$.

On peut aller plus vite en déléguant les calculs à Xcas :

```
/* parabole et cercle, version élève */
A:=point(xa,ya):; B:=point(xb,yb):;
equation(perpen_bisector(A,B))
M:=unapply((xa^2-xb^2+ya^2-yb^2)/(2*ya-2*yb)+x*(-xa+xb)/(ya-yb),[xa,ya,xb,yb,x])
/* equation de la médiatrice de (AB) */
```

```
eq1:=M(2,4,-4,16,x)
/* equation de la médiatrice de (AC) */
eq2:=M(2,4,3,9,x)
/* coordonnées du point d'intersection des médiatrices */
[xw,yw]:=solve([y=eq1,y=eq2],[x,y])[0]
/* equation du cercle de centre W passant par A */
eq3:=equation(circle(point(xw,yw),distance(point(2,4),point(xw,yw))))
/* Liste des abscisses des intersections cercle-parabole */
Labs:=solve(subst(eq3,y=x^2),x)
somme(Labs)

```